

SCIENCE & VIE

MENSUEL

N° 850 JUILLET 1988

Pluton
au plus près
de nous

Les parfums
messages
secrets

**ANTI-CAMBRIOLE:
LA MEILLEURE
TECHNIQUE**


M 2578 - 850 - 17,00 F



3792578017005 08500

Le lac des tortues

INFORMATIQUE AMUSANTE



Nous vous proposons, ce mois-ci, un jeu d'arcades. Nous l'avons appelé le Lac des tortues. L'action se déroule en pleine jungle. Sur un lac poussent des nénuphars mais, pour notre grenouille, que vous guiderez, ceux de couleur rose sont les plus propices à la découverte des insectes dont elle se réglera.

Malheureusement ces derniers ne poussent que sur le sable fin des berges du lac, et la plage sur laquelle se trouve notre grenouille en est dépourvue. Il lui faut donc traverser le lac. Cependant la chose n'est pas simple car, d'un type un peu spécial, notre grenouille ne sait pas nager. Sa traversée devra donc s'effectuer en sautant de feuille en feuille sur les nénuphars flottant sur le lac, puis sur le dos de tortues batifolant dans les eaux sombres.

Cependant les problèmes ne s'arrêtent pas là. En effet, les nénuphars, diaphanes et d'une extrême fragilité, ne peuvent servir que de tremplin aux bonds de la grenouille. Si elle y séjourne trop longtemps, la feuille coule et notre grenouille avec. De même, il n'est possible de sauter sur le dos d'une tortue que si elle se trouve exactement en face de la grenouille. Qui plus est, les tortues, d'humeur folâtre, aiment plonger de manière imprévisible. Pour terminer, un temps maximum sera imparti pour la traversée. A vous donc d'aider notre pauvre grenouille à traverser le lac des tortues en un minimum de temps.

Ces quelques règles fixées, passons maintenant à l'écriture du programme. En premier lieu, nous demanderons à l'Amstrad de pas-


```

10 MODE°O:CLS:CLEAR
20 REM *****
30 REM *
40 REM * DETERMINATION DES COULEURS DU FOND. *
50 REM *
60 REM *****
70 BORDER 9:PAPER 5:CLS
80 REM *****
90 REM *
100 REM * DEFFINITION DES SYMBOLES GRAPHIQUES *
110 REM *
120 REM *****
130 SYMBOL AFTER 127
140 SYMBOL 128,8,12,9,15,31,31,95,255
150 SYMBOL 129,0,4,206,252,252,254,254,255
160 SYMBOL 130,255,95,31,31,15,9,12,6
170 SYMBOL 131,255,254,254,252,252,206,4,0
180 SYMBOL 132,0,32,115,63,63,127,127,255
190 SYMBOL 133,96,48,144,240,248,248,250,255
200 SYMBOL 134,255,127,127,63,63,115,32,0
210 SYMBOL 135,255,250,248,248,240,144,48,96
220 SYMBOL 136,3,28,32,67,70,72,153,146
230 SYMBOL 137,192,56,4,194,98,18,153,73
240 SYMBOL 138,146,153,72,70,67,32,28,3
250 SYMBOL 139,73,153,188,98,194,4,56,192
260 SYMBOL 140,24,153,189,153,189,189,255,126
270 SYMBOL 141,126,126,60,153,219,219,126,60
280 SYMBOL 142,1,7,31,63,127,255,255,255
290 SYMBOL 143,240,252,254,255,255,255,255,255
300 SYMBOL 144,255,255,127,127,63,63,31,15
310 SYMBOL 145,255,127,76,96,160,176,152,8
320 REM *****
330 REM *
340 REM * TABLEAU MEMORISATION TROUS NENUPHARES *
350 REM *
360 REM *****
370 DIM E(7,3)
380 REM *****
390 REM *
400 REM * INITIALISATION DES VARIABLES *
410 REM *
420 REM *****
430 LET PGH=10:PGV=23:LET PT=0
440 LET DMH=4:DMV=4
450 LET FIN=0:LET TS=240
460 LET GH=CHR$(140)
470 LET GB=CHR$(141)
480 LET THD$=" "+CHR$(132)+CHR$(133)
490 LET TBD$=" "+CHR$(134)+CHR$(135)
500 LET THG$=CHR$(128)+CHR$(129)+" "
510 LET TBG$=CHR$(130)+CHR$(131)+" "
520 LET EAUH$=" "+CHR$(136)+CHR$(137)+" "
530 LET EHS$=CHR$(136)+CHR$(137)
540 LET EAUB$=" "+CHR$(138)+CHR$(139)+" "
550 LET EBS$=CHR$(138)+CHR$(139)
560 LET EF$=" "
570 LET NH$=CHR$(142)+CHR$(143)
580 LET NB$=CHR$(144)+CHR$(145)

590 REM *****
600 REM *
610 REM * MISE EN PLACE DES NENUPHARES *
620 REM *
630 REM *****
640 PEN 12
650 FOR V=14 TO 20 STEP 3
660 FOR H=1 TO 19 STEP 3
670 LOCATE H,V:PRINT NH$
680 LOCATE H,V+1:PRINT NB$
690 NEXT H
700 NEXT V

710 REM *****
720 REM *
730 REM * MISE EN PLACE DES "PLAGES" *
740 REM *
750 REM *****
760 WINDOW #1,1,20,1,4:PAPER #1,1:CLS#1
770 WINDOW #2,1,20,23,25:PAPER #2,1:CLS#2
780 PEN #1,11
790 FOR I=1 TO 19 STEP 3
800 LOCATE #1,I,2:PRINT #1,NH$
810 LOCATE #1,I,3:PRINT #1,NB$
820 NEXT I

830 REM *****
840 REM *
850 REM * CONDITIONS DE DEPART DES TORTUES *
860 REM *
870 REM *****
880 LET DA=1:DB=-1:DC=1:DD=-1
890 LET XA=5:XB=7:XC=12:XD=15
900 LET AA=0:AB=0:AC=0:AD=0
910 REM *****
920 REM *
930 REM * EFFACEMENT DES NENUPHARES *
940 REM *
950 REM *****
960 FOR I=1 TO 10
970 LET X=INT(RND*7)+1
980 LET Y=INT(RND*3)+1
990 LET E(X,Y)=1
1000 LOCATE ((X-1)*3)+1,((Y-1)*3)+14:PRINT " "
1010 LOCATE ((X-1)*3)+1,((Y-1)*3)+15:PRINT " "
1020 NEXT I
1030 REM *****
1040 REM *
1050 REM * PLONGEONS ALEATOIRES DES TORTUES *
1060 REM *
1070 REM *****
1080 LET PL=INT(RND*50)
1090 REM *****
1100 REM *
1110 REM * AFFICHAGE DU TEMPS DE JEU IMPARTI *
1120 REM *
1130 REM *****
1140 LOCATE #1,1,1: PEN #1,3:PRINT #1,"TEMPS:TS
1150 REM *****
1160 REM *
1170 REM * ANIMATION DE LA PREMIERE TORTUE *
1180 REM *
1190 REM *****
1200 LOCATE XA,5
1210 IF PL=1 THEN LET AA=0
1220 IF DA=1 AND PL<>1 AND AA>5 THEN PEN 9:PRINT T
HD$
1230 IF DA=-1 AND PL<>1 AND AA>5 THEN PEN 9:PRINT
THG$
1240 IF PL=1 THEN PEN 2:PRINT EAUH$
1250 IF AA=1 THEN PRINT EF$
1260 LOCATE XA,6
1270 IF DA=1 AND PL<>1 AND AA>5 THEN PEN 9:PRINT T
BD$
1280 IF DA=-1 AND PL<>1 AND AA>5 THEN PEN 9:PRINT
TBG$
1290 IF PL=1 THEN PEN 2:PRINT EAUB$
1300 IF AA=1 THEN PRINT EF$
1310 LET XA=XA+DA
1320 IF XA<2 OR XA>16 THEN LET DA=-DA
1330 LET AA=AA+1
1340 REM *****
1350 REM *
1360 REM * ANIMATION DE LA SECONDE TORTUE *
1370 REM *
1380 REM *****
1390 LOCATE XB,7
1400 IF PL=2 THEN LET AB=0
1410 IF DB=1 AND PL<>2 AND AB>5 THEN PEN 9:PRINT T
HD$
1420 IF DB=-1 AND PL<>2 AND AB>5 THEN PEN 9:PRINT
THG$
1430 IF PL=2 THEN PEN 2:PRINT EAUH$
1440 IF AB=1 THEN PRINT EF$
1450 LOCATE XB,8
1460 IF DB=1 AND PL<>2 AND AB>5 THEN PEN 9:PRINT T
BD$
1470 IF DB=-1 AND PL<>2 AND AB>5 THEN PEN 9:PRINT
TBG$
1480 IF PL=2 THEN PEN 2:PRINT EAUB$
1490 IF AB=1 THEN PRINT EF$
1500 LET XB=XB+DB
1510 IF XB<2 OR XB>16 THEN LET DB=-DB
1520 LET AB=AB+1
1530 REM *****
1540 REM *
1550 REM * ANIMATION DE LA TROISIEME TORTUE *
1560 REM *
1570 REM *****
1580 LOCATE XC,9
1590 IF PL=3 THEN LET AC=0

```



```

1600 IF DC=1 AND PL<>3 AND AC>5 THEN PEN 9:PRINT T
HD$
1610 IF DC=-1 AND PL<>3 AND AC>5 THEN PEN 9:PRINT
THG$
1620 IF PL=3 THEN PEN 2:PRINT EAUH$
1630 IF AC=1 THEN PRINT EF$
1640 LOCATE XC,10
1650 IF DC=1 AND PL<>3 AND AC>5 THEN PEN 9:PRINT T
BD$
1660 IF DC=-1 AND PL<>3 AND AC>5 THEN PEN 9:PRINT
TBG$
1670 IF PL=3 THEN PEN 2:PRINT EAUB$
1680 IF AC=1 THEN PRINT EF$
1690 LET XC=XC+DC
1700 IF XC<2 OR XC>16 THEN LET DC=-DC
1710 LET AC=AC+1
1720 REM *****
1730 REM *
1740 REM * ANIMATION DE LA QUATRIEME TORTUE *
1750 REM *
1760 REM *****
1770 LOCATE XD,11
1780 IF PL=4 THEN LET AD=0
1790 IF DD=1 AND PL<>4 AND AD>5 THEN PEN 9:PRINT T
HD$
1800 IF DD=-1 AND PL<>4 AND AD>5 THEN PEN 9:PRINT
THG$
1810 IF PL=4 THEN PEN 2:PRINT EAUH$
1820 IF AD=1 THEN PRINT EF$
1830 LOCATE XD,12
1840 IF DD=1 AND PL<>4 AND AD>5 THEN PEN 9:PRINT T
BD$
1850 IF DD=-1 AND PL<>4 AND AD>5 THEN PEN 9:PRINT
TBG$
1860 IF PL=4 THEN PEN 2:PRINT EAUB$
1870 IF AD=1 THEN PRINT EF$
1880 LET XD=XD+DD
1890 IF XD<2 OR XD>16 THEN LET DD=-DD
1900 LET AD=AD+1
1910 REM *****
1920 REM *
1930 REM * AIGUILLAGE VERS LES SOUS-ROUTINES *
1940 REM *
1950 REM *****
1960 IF TS=0 THEN LET FIN=1
1970 IF PGV>13 THEN GOSUB 2250
1980 IF PGV<13 THEN GOSUB 2510
1990 IF PGV<>3 AND FIN=0 THEN GOTO 1080
2000 IF PGV>4 THEN GOTO 2090
2010 PEN 13
2020 LOCATE PGH,3:PRINT GH$
2030 LOCATE PGH,4:PRINT GB$
2040 REM *****
2050 REM *
2060 REM * CONTROLE FIN DE JEU *
2070 REM *
2080 REM *****
2090 PEN 3
2100 LOCATE 7,14:PRINT "PERDU ! "
2110 IF FIN=0 THEN LOCATE 7,14:PRINT "BRAVO ! "
2120 LOCATE 7,15:PRINT " "
2130 LOCATE 1,17:PRINT "POUR UN AUTRE ESSAI "
2140 LOCATE 1,18:PRINT " "
2150 LOCATE 1,20:PRINT "TAPER LA TOUCHE 'P'."
2160 LOCATE 1,21:PRINT " "
2170 IF UPPER$(INKEY$)<>"P" THEN GOTO 2170
2180 GOTO 10
2190 REM *****
2200 REM *
2210 REM * SOUS-ROUTINE D'ANIMATION DE LA *
2220 REM * GRENOUILLE SUR FEUILLES DE NENUPHARES *
2230 REM *
2240 REM *****
2250 IF PGV=23 THEN PAPER 1:LOCATE PGH,PGV:PRINT E
F$:LOCATE PGH,PGV+1:PRINT EF$:PAPER 5
2260 LET TS=TS-1
2270 PEN 2
2280 IF PGV<23 THEN LOCATE PGH,PGV:PRINT EH$:LOCAT
E PGH,PGV+1:PRINT EB$
2290 IF DMV<4 THEN LET E(DMH,DMV)=1
2300 IF INKEY(1)=0 THEN LET PGH=PGH+3:LET DMH=DMH+
1
2310 IF PGH>19 THEN LET PGH=19:LET DMH=7
2320 IF INKEY(8)=0 THEN LET PGH=PGH-3:LET DMH=DMH-
1
2330 IF PGH<1 THEN LET PGH=1:LET DMH=1
2340 IF INKEY(2)=0 THEN LET PGV=PGV+3:LET DMV=DMV+
1
2350 IF PGV>24 THEN LET PGV=24:LET DMV=4
2360 IF INKEY(0)=0 THEN LET PGV=PGV-3:LET DMV=DMV-
1
2370 PEN 13
2380 LOCATE PGH,PGV:PRINT GH$
2390 LOCATE PGH,PGV+1:PRINT GB$
2400 IF DMV=4 THEN GOTO 2430
2410 IF E(DMH,DMV)=1 THEN LET FIN=1
2420 LET PT=1
2430 RETURN
2440 REM *****
2450 REM *
2460 REM * SOUS-ROUTINE D'ANIMATION DE LA *
2470 REM * GRENOUILLE SUR LE DOS DES TORTUES *
2480 REM *
2490 REM *****
2500 LET TS=TS-1
2510 IF PGV<>11 THEN GOTO 2670
2520 IF PL=4 OR AD<6 THEN GOTO 2630
2530 IF PGH<>XD AND PGH<>XD+1 AND PGH<>XD+2 THEN G
OTO 2630
2540 IF PT=1 THEN GOTO 2560
2550 IF INKEY(0)=0 THEN LET PGV=PGV-2
2560 PEN 13
2570 LET PT=0
2580 LOCATE XD+1,11:PRINT GH$
2590 LOCATE XD+1,12:PRINT GB$
2600 LET PGH=XD+1
2610 IF INKEY(2)=0 THEN LET PGV=PGV+3:LET DMH=INT(
PGH/3)+1:LET PGH=((DMH-1)*3)+1:LET DMV=1
2620 RETURN
2630 PEN 2:LOCATE PGH,11:PRINT EH$
2640 LOCATE PGH,12:PRINT EB$
2650 LET FIN=1
2660 RETURN
2670 IF PGV<>9 THEN GOTO 2810
2680 IF PL=3 OR AC<6 THEN GOTO 2770
2690 IF PGH<>XC AND PGH<>XC+1 AND PGH<>XC+2 THEN G
OTO 2770
2700 IF INKEY(2)=0 THEN LET PGV=PGV+2
2710 IF INKEY(0)=0 THEN LET PGV=PGV-2
2720 PEN 13
2730 LOCATE XC+1,9:PRINT GH$
2740 LOCATE XC+1,10:PRINT GB$
2750 LET PGH=XC+1
2760 RETURN
2770 PEN 2:LOCATE PGH,9:PRINT EH$
2780 LOCATE PGH,10:PRINT EB$
2790 LET FIN=1
2800 RETURN
2810 IF PGV<>7 THEN GOTO 2950
2820 IF PL=2 OR AB<6 THEN GOTO 2910
2830 IF PGH<>XB AND PGH<>XB+1 AND PGH<>XB+2 THEN G
OTO 2910
2840 IF INKEY(2)=0 THEN LET PGV=PGV+2
2850 IF INKEY(0)=0 THEN LET PGV=PGV-2
2860 PEN 13
2870 LOCATE XB+1,7:PRINT GH$
2880 LOCATE XB+1,8:PRINT GB$
2890 LET PGH=XB+1
2900 RETURN
2910 PEN 2:LOCATE PGH,7:PRINT EH$
2920 LOCATE PGH,8:PRINT EB$
2930 LET FIN=1
2940 RETURN
2950 IF PL=1 OR AA<6 THEN GOTO 3040
2960 IF PGH<>XA AND PGH<>XA+1 AND PGH<>XA+2 THEN G
OTO 3040
2970 IF INKEY(2)=0 THEN LET PGV=PGV+2
2980 IF INKEY(0)=0 THEN LET PGV=PGV-2
2990 PEN 13
3000 LOCATE XA+1,5:PRINT GH$
3010 LOCATE XA+1,6:PRINT GB$
3020 LET PGH=XA+1
3030 RETURN
3040 PEN 2:LOCATE PGH,5:PRINT EH$
3050 LOCATE PGH,6:PRINT EB$
3060 LET FIN=1
3070 RETURN

```


ser en mode 0 ; ceci afin de profiter de nombreuses couleurs. Nous déterminerons ensuite la couleur du fond et du tour de l'écran.

Nous utiliserons de nombreux symboles graphiques. Nous les redéfinirons des lignes 130 à 310. Un tableau de mémorisation sera créé à la ligne 370. Son but sera d'indiquer au programme, lors du déroulement du jeu, la position des feuilles de nénuphars flottant encore. Nous utiliserons également un grand nombre de variables ; elles seront initialisées des lignes 430 à 580. Notons que toutes ces variables ne concernent pas des paramètres du jeu. En effet, comme nous devons animer simultanément de nombreux symboles graphiques, nous les avons regroupés par "familles" sous la forme d'une variable. Ainsi TDH\$ permettra d'afficher directement une demi-tortue. Il en sera de même pour toutes les autres variables se terminant par le signe \$. Précisons que cette opération nous permettra, d'une part, de simplifier la frappe des lignes suivantes du programme et, d'autre part, améliorera sa vitesse d'exécution ; ce qui est toujours recherché dans un jeu d'animation.

Ceci terminé, nous préparerons l'écran de jeu en mettant en place les nénuphars grâce à deux boucles FOR... NEXT imbriquées. Cette tâche sera effectuée par les lignes 640 à 700. De même les "plages" seront déterminées par deux fenêtres — lignes 760 et 770 — dont nous modifierons la couleur par rapport au reste de l'écran. Une nouvelle série de variables — lignes 880 à 900 — nous permettra de fixer les conditions de départ des tortues ; position et sens de déplacement.

Dernière opération avant le début du jeu : effacer certains des nénuphars, ceci afin de corser le jeu. Cette opération sera effectuée de manière aléatoire, par les lignes 960 à 1020. Le jeu pourra donc débiter.

En premier lieu la variable PL nous permettra de faire plonger de manière imprévisible nos tortues. Nous afficherons le crédit de temps disponible pour la partie — ligne 1140 — et l'animation des tortues sera assurée.

Comme nous disposons de quatre tortues, nous trouverons quatre

animations distinctes. Cependant leur principe de fonctionnement est identique. Nous n'examinerons donc que la première. En premier lieu, nous trouverons une commande de position d'affichage de la tortue — instruction LOCATE de la ligne 1200. XA correspondra à sa position horizontale et sera donc une variable commandée par l'animation.

Le second paramètre sera, par contre, fixe — dans ce cas 5, puisque les tortues ne se déplacent qu'horizontalement.

Nous prendrons ensuite en compte la variable PL, ceci afin de savoir si la tortue est en train de nager ou si elle a plongé. Enfin, la valeur de DA — variable fixant la direction de déplacement — sera contrôlée de manière à déterminer quel symbole graphique utiliser pour éviter que notre tortue semble nager à reculons. Notons que AA est une variable auxiliaire permettant de simuler le trajet de la tortue sous l'eau après le plongeon.

Cette première phase ne concerne que la moitié supérieure du dessin de la tortue, une seconde concernera la moitié basse. Enfin, en ligne 1320, nous modifierons, ou non, le signe de la variable DA si la tortue a atteint les limites de l'écran. Cette opération provoquera l'inversion du sens de son déplacement.

Comme nous l'avons dit plus haut, nous trouverons donc quatre séquences de programmes très proches ; une par tortue. Seule une modification des variables sera à signaler. XA deviendra XB puis XC et XD, AA, AB, AC, et AD, etc. Cette animation terminée, il nous faudra également contrôler les déplacements de notre grenouille. Ceci sera effectué par l'intermédiaire de deux sous-routines que nous étudierons plus loin. C'est pour cela que nous trouverons un aiguillage vers celles-ci, placé des lignes 1970 à 2030.

Notons que nous aiguillerons également ici éventuellement le programme, soit sur un rebouclage, destiné à poursuivre le jeu, soit sur sa fin. Dans ce dernier cas, il se rendra en ligne 2090. Là, en fonction de la valeur prise par la variable FIN, soit "PERDU !" soit "BRAVO !" sera affiché sur l'écran. Le fait de taper "P" au clavier per-

mettra, quel que soit le cas, de reprendre le jeu. Restent donc les deux sous-routines.

La première, placée des lignes 2250 à 2430, concerne les sauts de la grenouille sur les feuilles de nénuphars. Nous utiliserons donc régulièrement la fonction INKEY(x), où x représente le numéro de la touche déterminé par Amstrad, pour savoir, en fonction de la touche frappée quelle est la direction choisie par le joueur pour le saut de la grenouille. Précisons que cette pratique nous permet, contrairement à INKEY\$ également fréquemment utilisée dans nos programmes, de pouvoir "sauter en biais" en appuyant sur deux touches simultanément. Nous ferons également appel au tableau E pour contrôler régulièrement si le résultat du saut ne conduit pas à la noyade de notre grenouille — ligne 2410 — auquel cas nous donnerions à la variable FIN la valeur 1.

La seconde sous-routine, placée des lignes 2500 à 3070 concerne les déplacements de la grenouille sur le dos des tortues. Ici nous ferons également fréquemment appel à l'instruction INKEY(x) mais x ne pourra valoir que 0 ou 2 ; il sera effectivement interdit de sauter en biais. Comme nous avons quatre tortues, cette sous-routine comportera également quatre parties dont le fonctionnement sera tout à fait comparable.

La frappe de ce programme ne doit pas poser de problème particulier. De même son utilisation est des plus simples. Après l'avoir entièrement tapé, RUN sera demandé. Aussitôt l'écran de jeu s'affichera et le chronométrage débutera. Rappelons que les seules zones de sécurité sont les plages.

Vous aurez donc tout loisir de préparer, dans la limite du temps imparti, votre stratégie avant de vous lancer sur les feuilles de nénuphars. Signalons encore que chaque feuille ne peut être utilisée qu'une fois en tant que tremplin. Impossible donc de rester sur une même feuille ou de tenter de réutiliser une feuille ayant déjà sombré. Seul élément aléatoire du jeu : le plongeon des tortues. A vous donc de l'éviter en ne restant pas trop longtemps sur la même tortue, quitte à regagner d'urgence la berge de départ.